
1. Overview

This document lists all changes needed to correct problems in the TTCN implementation of test case 7.3.4.4_5s which is part of the POS test suite ‘POS_13wk08’ in the ‘iwd-EUTRA-B2012-03_D13wk08’ delivery.
The test case can be demonstrated to run with two LTE UEs (see section 5). Execution logs are provided as evidence.

3. Verification Test Summary

Test Case:
TC_7_3_4_4_5s
Test Group:
Positioning\EUTRA_Positioning_Testcases.ttcn
ATS Version:
iwd-EUTRA-B2012-03_D13wk08
System Simulator used:
R&S CMW500 
UE used:
Qualcomm MDM 8960           and
Renesas Mobile MP5232-1 
Verification Status:
PASS


4. Corrections required for POS common modules
See change 4.1 of R5s130157 (TC 7.3.4.2_5s) plus the following additional changes:
Change 1 – Extension of template ‘cs_LppAcknowledgement()’ 
	Template name
	cs_LppAcknowledgement()

	Reason for change
	Parameter “p_endTransaction” is hardcoded to value ‘true’ in template “cs_LppAcknowledgement”. But in order to handle the optional ACK in step 4 of TC function ‘f_TC_7_3_4_4_5s_EUTRA(), it is required to set “p_endTransaction” also to value ’false’.

	Summary of change
	Extended template “cs_LppAcknowledgement” with parameter “p_endTransaction”.

	TTCN module
	Positioning\EUTRA_Positioning_Templates.ttcn

	MCC160 Comment
	


Before change

  //-----------------------------------------------------------------------------------------------------------------------------
  function fl_SetSupportedECIDMeas () return bitstring
  {
    var bitstring v_ECIDMeas := '000'B;
    if (pc_ECID_Rsrp) {
      v_ECIDMeas := replace (v_ECIDMeas, 0, 1, '1'B); // Set first bit to 1
    }
    if (pc_ECID_Rsrq) {
      v_ECIDMeas := replace (v_ECIDMeas, 1, 1, '1'B); // Set second bit to 1
    }
    if (pc_ECID_UeRxTx) {
      v_ECIDMeas := replace (v_ECIDMeas, 2, 1, '1'B); // Set third bit to 1
    }
    return v_ECIDMeas;
  }
  template (value) LPP_Message cs_LppAcknowledgement (SequenceNumber p_SeqNumber) := {
    transactionID := omit,
    endTransaction := true,
    sequenceNumber := omit,
    acknowledgement := {
      ackRequested := false,
      ackIndicator := p_SeqNumber
    },
    lpp_MessageBody := omit
  };
  //----------------------------------------------------------------------------------
  // Templates to request assistance data
  // Info taken from 37.571-2, Table 5.4-2 and clause 5.4.1
After change

  //-----------------------------------------------------------------------------------------------------------------------------
  function fl_SetSupportedECIDMeas () return bitstring
  {
    var bitstring v_ECIDMeas := '000'B;
    if (pc_ECID_Rsrp) {
      v_ECIDMeas := replace (v_ECIDMeas, 0, 1, '1'B); // Set first bit to 1
    }
    if (pc_ECID_Rsrq) {
      v_ECIDMeas := replace (v_ECIDMeas, 1, 1, '1'B); // Set second bit to 1
    }
    if (pc_ECID_UeRxTx) {
      v_ECIDMeas := replace (v_ECIDMeas, 2, 1, '1'B); // Set third bit to 1
    }
    return v_ECIDMeas;
  }
  template (value) LPP_Message cs_LppAcknowledgement (SequenceNumber p_SeqNumber,
                                                      boolean  p_endTransaction := true) := {
    transactionID := omit,
    endTransaction := p_endTransaction,
    sequenceNumber := omit,
    acknowledgement := {
      ackRequested := false,
      ackIndicator := p_SeqNumber
    },
    lpp_MessageBody := omit
  };
  //----------------------------------------------------------------------------------
  // Templates to request assistance data
  // Info taken from 37.571-2, Table 5.4-2 and clause 5.4.1
Change 2 – Extension of function ‘f_POS_SendLppAck()’ 
	Function name
	f_POS_SendLppAck()

	Reason for change
	Function ‘f_POS_SendLppAck()’ is extended to consider non-default value of parameter “p_endTransaction”.

	Summary of change
	Extended function ‘f_POS_SendLppAck()’ with parameter “p_endTransaction”.

	TTCN module
	Positioning\EUTRA_Positioning_Functions.ttcn

	MCC160 Comment
	


Before change

  // To send LPP ACKNOWLEDGEMENT if requested by the UE
  function f_POS_SendLppAck(EUTRA_CellId_Type p_CellId,
                            octetstring p_RoutingId,
                            LPP_Message p_LppMsgFromUE) runs on EUTRA_PTC
  {
    var octetstring v_EncodedLppMessage := f_POS_EncodeLppMessage(cs_LppAcknowledgement(p_LppMsgFromUE.sequenceNumber));
    if (isvalue(p_LppMsgFromUE.acknowledgement)) {// UE requested ack
      SRB.send(cas_SRB_NasPdu_REQ(p_CellId,
                                  tsc_SRB2,
                                  cs_TimingInfo_Now,
                                  cs_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered,
                                                 cs_DL_GENERIC_NAS_TRANSPORT(tsc_LPP_MessageContainer,
                                                                             cs_GenericMsgContainer(v_EncodedLppMessage),
                                                                             cs_RoutingInfo(p_RoutingId)))));
    }
  }
After change

  // To send LPP ACKNOWLEDGEMENT if requested by the UE
  function f_POS_SendLppAck(EUTRA_CellId_Type p_CellId,
                            octetstring p_RoutingId,
                            LPP_Message p_LppMsgFromUE,
                            boolean  p_endTransaction := true) runs on EUTRA_PTC
  {
    var octetstring v_EncodedLppMessage := f_POS_EncodeLppMessage(cs_LppAcknowledgement(p_LppMsgFromUE.sequenceNumber, p_endTransaction));
    if (isvalue(p_LppMsgFromUE.acknowledgement)) {// UE requested ack
      SRB.send(cas_SRB_NasPdu_REQ(p_CellId,
                                  tsc_SRB2,
                                  cs_TimingInfo_Now,
                                  cs_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered,
                                                 cs_DL_GENERIC_NAS_TRANSPORT(tsc_LPP_MessageContainer,
                                                                             cs_GenericMsgContainer(v_EncodedLppMessage),
                                                                             cs_RoutingInfo(p_RoutingId)))));
    }
  }

5. Corrections required for test case 7.3.4.4_5s
Change 1 – Correction to function ‘f_TC_7_3_4_4_5s_EUTRA()’ 
	Function name
	f_TC_7_3_4_4_5s_EUTRA()

	Reason for change
	1. EUTRA initialization is done with SIB combination no.1 (i.e. without SIB4) even though TC 7.3.4.4_5s simulates an intra-frequency multi cell scenario.

2.   It is not ensured that the UE has successfully detected neighboring cells 2 and 4 when starting with OTDOA procedures in the test body.

3.   Similar to the LPP REQUEST ASSISTANCE DATA message in step 2, the LPP REQUEST ASSISTANCE DATA message sent from the SS in step 3 should contains routing info. This routing info is expected to be echoed in the LPP PROVIDE ASSISTANCE DATA message sent by the UE in step 4.

4.   The behavior in step 4 does not consider that a UE may request to send an ACK in step 3.

Moreover, the error checking in step 4 covers only the checking if IE “otdoa_ProvideLocationInformation” is present instead of checking also for errors reported in this IE.

	Summary of change
	1.   Called function f_EUTRA_Init() with parameter ‘c2’ instead of ‘c1’ at the very beginning. As a consequence, function f_EUTRA_CellInfo_SetSIB4 had to be additionally called for cell 1, 2 and 4.

See also introduction of templates in section 5 change 2 of R5s130157.

2.   Cells 2 and 4 are switched on with a defined power level at the end of the preamble and a delay of 15 sec. was introduced to ensure that the UE successfully detected cell 2 and 4 when starting with the test body.

3.   Routing info considered in the send template in step 3 as well as in the receive template in step 4.

4.   Optional sending of ACK was added to step 4.

Moreover, the error checking in step 4 was extended. 

	TTCN module
	Positioning\EUTRA_Positioning_Testcases.ttcn

	MCC160 Comment
	


Before change

  /*
   * @desc      REFERENCE TS 37.571-2 clause 7.3.4.4, Subtest 5 (UE supporting OTDOA)
   * @status    NOT APPROVED
   */
  function f_TC_7_3_4_4_5s_EUTRA ( ) runs on EUTRA_POS_PTC {
    const TransactionNumber tsc_TransactionNumber := 0; // Any value between 0 and 255.
    const octetstring tsc_RoutingId := '01234567'O; // Selected by the network, use arbitrary value.
    var SRB_COMMON_IND v_ReceivedAsp;
    var LPP_Message v_LppMessage;
    var octetstring v_Octetstring;
    var octetstring v_EncodedLppMessage;
    var TransactionNumber v_TransactionNumber := 0; // To store Transaction number sent by the UE in LPP Request Assistance Data
    var PRS_Info v_PrsInfo := valueof (cs_PrsInfo_Def);
    var PRS_Info v_PrsInfo_Tmp;
    var CellIdentity v_CellIdentity1;
    var CellIdentity v_CellIdentity2;
    var CellIdentity v_CellIdentity4;
    var ECGI v_Ecgi1;
    var ECGI v_Ecgi2;
    var ECGI v_Ecgi4;
    var NAS_PlmnId v_PLMN1_Oct;
    var NAS_PlmnId v_PLMN2_Oct;
    var NAS_PlmnId v_PLMN4_Oct;
    f_EUTRA_Init (c1);
    // create and configure cells for OTDOA
    f_POS_EUTRA_CellConfig_OTDOA (eutra_Cell1);
    f_POS_EUTRA_CellConfig_OTDOA (eutra_Cell2);
    f_POS_EUTRA_CellConfig_OTDOA (eutra_Cell4);
    // Preamble:
    //   - UE is in state Generic RB established (state 3) on Cell 1
    // ==========================================================================
    f_EUTRA_Preamble(eutra_Cell1, STATE2_IDLEUPDATE);
    f_EUTRA_RbEst_Def(eutra_Cell1);
    f_EUTRA_TestBody_Set(true);
    //@siclog "Step 1a" siclog@
    //Clear stored OTDOA assistance data
    SRB.send(cas_SRB_NasPdu_REQ(eutra_Cell1,
                                tsc_SRB2,
                                cs_TimingInfo_Now,
                                cs_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered,
                                               cs_EUTRA_RESETUEPOSITIONINGSTOREDINFORMATION (tsc_OtdoaUEPosTech))));
    //@siclog "Step 2" siclog@
    //SS sends LPP REQUEST LOCATION INFORMATION
    v_EncodedLppMessage := f_POS_EncodeLppMessage (cs_LppRequestLocationInfo (tsc_TransactionNumber,
                                                                          locationMeasurementsRequired,
                                                                          tsc_ResponseTimeDef,
                                                                          omit,
                                                                          cs_OtdoaRequestLocationInfo (true), //@sic R5-130112 sic@
                                                                          omit));
    SRB.send(cas_SRB_NasPdu_REQ(eutra_Cell1,
                                tsc_SRB2,
                                cs_TimingInfo_Now,
                                cs_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered,
                                               cs_DL_GENERIC_NAS_TRANSPORT(tsc_LPP_MessageContainer,
                                                                           cs_GenericMsgContainer(v_EncodedLppMessage),
                                                                           cs_RoutingInfo (tsc_RoutingId)))));
    //@siclog "Step 3" siclog@
    //Receive LPP Request Assistance Data
    SRB.receive(car_SRB2_NasPdu_IND(eutra_Cell1,
                                    cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered,
                                                      cr_UL_GENERIC_NAS_TRANSPORT(tsc_LPP_MessageContainer,
                                                                                  omit))))
    -> value v_ReceivedAsp;
    v_Octetstring := v_ReceivedAsp.Signalling.Nas[0].Pdu.Msg.ul_GENERIC_NAS_TRANSPORT.genericMessageContainer.genericMsgContainerValue;
    v_LppMessage := f_POS_LppMessage_DecodeAndMatch (v_Octetstring, cr_LppReqAssistanceDataOtdoa_Ack(f_EUTRA_CellInfo_GetPhyCellId(eutra_Cell1)));
    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Test Case 7.3.4.4 Step 3");
    // Get Transaction number sent by the UE
    v_TransactionNumber := v_LppMessage.transactionID.transactionNumber;
    //Set values needed for the OTDOA template
    v_PrsInfo_Tmp := f_POS_Set_PrsBandwidth (eutra_Cell1);
    v_PrsInfo.prs_Bandwidth := v_PrsInfo_Tmp.prs_Bandwidth;
    v_PLMN1_Oct := f_Asn2Nas_PlmnId (f_EUTRA_CellInfo_GetGutiPLMN (eutra_Cell1));
    v_PLMN2_Oct := f_Asn2Nas_PlmnId (f_EUTRA_CellInfo_GetGutiPLMN (eutra_Cell2));
    v_PLMN4_Oct := f_Asn2Nas_PlmnId (f_EUTRA_CellInfo_GetGutiPLMN (eutra_Cell4));
    v_Ecgi1 := f_POS_NAS_PlmnId2EcgiMccMnc (v_PLMN1_Oct);
    v_Ecgi2 := f_POS_NAS_PlmnId2EcgiMccMnc (v_PLMN2_Oct);
    v_Ecgi4 := f_POS_NAS_PlmnId2EcgiMccMnc (v_PLMN4_Oct);
    v_CellIdentity1 := f_EUTRA_CellInfo_GetCellIdentity (eutra_Cell1);
    v_CellIdentity2 := f_EUTRA_CellInfo_GetCellIdentity (eutra_Cell2);
    v_CellIdentity4 := f_EUTRA_CellInfo_GetCellIdentity (eutra_Cell4);
    v_Ecgi1.cellidentity := v_CellIdentity1;
    v_Ecgi2.cellidentity := v_CellIdentity2;
    v_Ecgi4.cellidentity := v_CellIdentity4;
    //@siclog "Step 4" siclog@
    //SS sends LPP PROVIDE ASSISTANCE DATA
    if (isvalue(v_LppMessage.acknowledgement)) {// UE requested ack
      v_EncodedLppMessage := f_POS_EncodeLppMessage (cs_LppProvideAssistData (v_TransactionNumber,
                                                                          targetDevice,
                                                                          omit,
                                                                          cs_OtdoaProvideAssistanceData(v_Ecgi1, v_PrsInfo, v_Ecgi2, v_Ecgi4),
                                                                          v_LppMessage.sequenceNumber));
      SRB.send(cas_SRB_NasPdu_REQ(eutra_Cell1,
                                  tsc_SRB2,
                                  cs_TimingInfo_Now,
                                  cs_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered,
                                                 cs_DL_GENERIC_NAS_TRANSPORT(tsc_LPP_MessageContainer,
                                                                             cs_GenericMsgContainer(v_EncodedLppMessage),
                                                                             cs_RoutingInfo (tsc_RoutingId)))));
    }
    else {// UE did not request ack
      v_EncodedLppMessage := f_POS_EncodeLppMessage (cs_LppProvideAssistData (v_TransactionNumber,
                                                                          targetDevice,
                                                                          omit,
                                                                          cs_OtdoaProvideAssistanceData(v_Ecgi1, v_PrsInfo, v_Ecgi2, v_Ecgi4)));
      SRB.send(cas_SRB_NasPdu_REQ(eutra_Cell1,
                                  tsc_SRB2,
                                  cs_TimingInfo_Now,
                                  cs_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered,
                                                 cs_DL_GENERIC_NAS_TRANSPORT(tsc_LPP_MessageContainer,
                                                                             cs_GenericMsgContainer(v_EncodedLppMessage),
                                                                             cs_RoutingInfo (tsc_RoutingId)))));
    }
    //@siclog "Step 5" siclog@
    //Receive LPP PROVIDE LOCATION INFORMATION
    SRB.receive(car_SRB2_NasPdu_IND(eutra_Cell1,
                                    cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered,
                                                      cr_UL_GENERIC_NAS_TRANSPORT(tsc_LPP_MessageContainer,
                                                                                  omit))))
    -> value v_ReceivedAsp;
    v_Octetstring := v_ReceivedAsp.Signalling.Nas[0].Pdu.Msg.ul_GENERIC_NAS_TRANSPORT.genericMessageContainer.genericMsgContainerValue;
    v_LppMessage := f_POS_LppMessage_DecodeAndMatch (v_Octetstring, cr_LppProvideLocationInformation_UEA (tsc_TransactionNumber));
    if (isvalue (v_LppMessage.lpp_MessageBody.c1.provideLocationInformation.criticalExtensions.c1.provideLocationInformation_r9.otdoa_ProvideLocationInformation)) {
      f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Test Case 7.3.4.4 Step 5");
    }
    else {
      f_EUTRA_SetVerdictFailOrInconc (__FILE__, __LINE__, "OTDOA Location Information not present");
    }
    //@siclog "Step 5a" siclog@
    //SS sends LPP ACKNOWLEDGEMENT if requested by the UE at step 5
    f_POS_SendLppAck (eutra_Cell1,
                      tsc_RoutingId,
                      v_LppMessage);
    // Postamble
    f_EUTRA_TestBody_Set(false);
    f_EUTRA_Postamble(eutra_Cell1, E2_CONNECTED);//@sic R5-103875 sic@
  };
After change

  /*
   * @desc      REFERENCE TS 37.571-2 clause 7.3.4.4, Subtest 5 (UE supporting OTDOA)
   * @status    NOT APPROVED
   */
  function f_TC_7_3_4_4_5s_EUTRA ( ) runs on EUTRA_POS_PTC {
    const TransactionNumber tsc_TransactionNumber := 0; // Any value between 0 and 255.
    const octetstring tsc_RoutingId := '01234567'O; // Selected by the network, use arbitrary value.
    var SRB_COMMON_IND v_ReceivedAsp;
    var LPP_Message v_LppMessage;
    var octetstring v_Octetstring;
    var octetstring v_EncodedLppMessage;
    var TransactionNumber v_TransactionNumber := 0; // To store Transaction number sent by the UE in LPP Request Assistance Data
    var PRS_Info v_PrsInfo := valueof (cs_PrsInfo_Def);
    var PRS_Info v_PrsInfo_Tmp;
    var CellIdentity v_CellIdentity1;
    var CellIdentity v_CellIdentity2;
    var CellIdentity v_CellIdentity4;
    var ECGI v_Ecgi1;
    var ECGI v_Ecgi2;
    var ECGI v_Ecgi4;
    var NAS_PlmnId v_PLMN1_Oct;
    var NAS_PlmnId v_PLMN2_Oct;
    var NAS_PlmnId v_PLMN4_Oct;
    f_EUTRA_Init (c2);
    f_EUTRA_CellInfo_SetSIB4 (eutra_Cell1, Cell1_DefSIB4_IntraFreqCellList);                                                                        
    f_EUTRA_CellInfo_SetSIB4 (eutra_Cell2, Cell2_DefSIB4_IntraFreqCellList);    
    f_EUTRA_CellInfo_SetSIB4 (eutra_Cell4, Cell4_DefSIB4_IntraFreqCellList);
    // create and configure cells for OTDOA
    f_POS_EUTRA_CellConfig_OTDOA (eutra_Cell1);
    f_POS_EUTRA_CellConfig_OTDOA (eutra_Cell2);
    f_POS_EUTRA_CellConfig_OTDOA (eutra_Cell4);
    // Preamble:
    //   - UE is in state Generic RB established (state 3) on Cell 1
    // ==========================================================================
    f_EUTRA_Preamble(eutra_Cell1, STATE2_IDLEUPDATE);    
    f_EUTRA_SetCellPower(eutra_Cell2, tsc_SuitableCellRS_EPRE);
    f_EUTRA_SetCellPower(eutra_Cell4, tsc_SuitableCellRS_EPRE);
    f_Delay (15.0);
    f_EUTRA_RbEst_Def(eutra_Cell1);
    f_EUTRA_TestBody_Set(true);
    //@siclog "Step 1a" siclog@
    //Clear stored OTDOA assistance data
    SRB.send(cas_SRB_NasPdu_REQ(eutra_Cell1,
                                tsc_SRB2,
                                cs_TimingInfo_Now,
                                cs_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered,
                                               cs_EUTRA_RESETUEPOSITIONINGSTOREDINFORMATION (tsc_OtdoaUEPosTech))));
    //@siclog "Step 2" siclog@
    //SS sends LPP REQUEST LOCATION INFORMATION
    v_EncodedLppMessage := f_POS_EncodeLppMessage (cs_LppRequestLocationInfo (tsc_TransactionNumber,
                                                                          locationMeasurementsRequired,
                                                                          tsc_ResponseTimeDef,
                                                                          omit,
                                                                          cs_OtdoaRequestLocationInfo (true), //@sic R5-130112 sic@
                                                                          omit));
    SRB.send(cas_SRB_NasPdu_REQ(eutra_Cell1,
                                tsc_SRB2,
                                cs_TimingInfo_Now,
                                cs_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered,
                                               cs_DL_GENERIC_NAS_TRANSPORT(tsc_LPP_MessageContainer,
                                                                           cs_GenericMsgContainer(v_EncodedLppMessage),
                                                                           cs_RoutingInfo (tsc_RoutingId)))));
    //@siclog "Step 3" siclog@
    //Receive LPP Request Assistance Data
    SRB.receive(car_SRB2_NasPdu_IND(eutra_Cell1,
                                    cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered,
                                                      cr_UL_GENERIC_NAS_TRANSPORT(tsc_LPP_MessageContainer,
                                                                                  cs_RoutingInfo (tsc_RoutingId)))))
    -> value v_ReceivedAsp;
    v_Octetstring := v_ReceivedAsp.Signalling.Nas[0].Pdu.Msg.ul_GENERIC_NAS_TRANSPORT.genericMessageContainer.genericMsgContainerValue;
    v_LppMessage := f_POS_LppMessage_DecodeAndMatch (v_Octetstring, cr_LppReqAssistanceDataOtdoa_Ack(f_EUTRA_CellInfo_GetPhyCellId(eutra_Cell1)));
    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Test Case 7.3.4.4 Step 3");
    // Get Transaction number sent by the UE
    v_TransactionNumber := v_LppMessage.transactionID.transactionNumber;
    //Set values needed for the OTDOA template
    v_PrsInfo_Tmp := f_POS_Set_PrsBandwidth (eutra_Cell1);
    v_PrsInfo.prs_Bandwidth := v_PrsInfo_Tmp.prs_Bandwidth;
    v_PLMN1_Oct := f_Asn2Nas_PlmnId (f_EUTRA_CellInfo_GetGutiPLMN (eutra_Cell1));
    v_PLMN2_Oct := f_Asn2Nas_PlmnId (f_EUTRA_CellInfo_GetGutiPLMN (eutra_Cell2));
    v_PLMN4_Oct := f_Asn2Nas_PlmnId (f_EUTRA_CellInfo_GetGutiPLMN (eutra_Cell4));
    v_Ecgi1 := f_POS_NAS_PlmnId2EcgiMccMnc (v_PLMN1_Oct);
    v_Ecgi2 := f_POS_NAS_PlmnId2EcgiMccMnc (v_PLMN2_Oct);
    v_Ecgi4 := f_POS_NAS_PlmnId2EcgiMccMnc (v_PLMN4_Oct);
    v_CellIdentity1 := f_EUTRA_CellInfo_GetCellIdentity (eutra_Cell1);
    v_CellIdentity2 := f_EUTRA_CellInfo_GetCellIdentity (eutra_Cell2);
    v_CellIdentity4 := f_EUTRA_CellInfo_GetCellIdentity (eutra_Cell4);
    v_Ecgi1.cellidentity := v_CellIdentity1;
    v_Ecgi2.cellidentity := v_CellIdentity2;
    v_Ecgi4.cellidentity := v_CellIdentity4;
    //@siclog "Step 4" siclog@
    //SS sends LPP PROVIDE ASSISTANCE DATA
    //SS sends LPP ACKNOWLEDGEMENT if requested by the UE at step 3
    f_POS_SendLppAck (eutra_Cell1, tsc_RoutingId, v_LppMessage, false);
    v_EncodedLppMessage := f_POS_EncodeLppMessage (cs_LppProvideAssistData (v_TransactionNumber,
                                                                            targetDevice,
                                                                            omit,
                                                                            cs_OtdoaProvideAssistanceData(v_Ecgi1, v_PrsInfo, v_Ecgi2, v_Ecgi4)));
    SRB.send(cas_SRB_NasPdu_REQ(eutra_Cell1,
                                tsc_SRB2,
                                cs_TimingInfo_Now,
                                cs_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered,
                                               cs_DL_GENERIC_NAS_TRANSPORT(tsc_LPP_MessageContainer,
                                                                           cs_GenericMsgContainer(v_EncodedLppMessage),
                                                                           cs_RoutingInfo (tsc_RoutingId)))));
    //@siclog "Step 5" siclog@
    //Receive LPP PROVIDE LOCATION INFORMATION
    SRB.receive(car_SRB2_NasPdu_IND(eutra_Cell1,
                                    cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered,
                                                      cr_UL_GENERIC_NAS_TRANSPORT(tsc_LPP_MessageContainer,
                                                                                  cs_RoutingInfo (tsc_RoutingId)))))
    -> value v_ReceivedAsp;
    v_Octetstring := v_ReceivedAsp.Signalling.Nas[0].Pdu.Msg.ul_GENERIC_NAS_TRANSPORT.genericMessageContainer.genericMsgContainerValue;
    v_LppMessage := f_POS_LppMessage_DecodeAndMatch (v_Octetstring, cr_LppProvideLocationInformation_UEA (tsc_TransactionNumber));
    if (isvalue (v_LppMessage.lpp_MessageBody.c1.provideLocationInformation.criticalExtensions.c1.provideLocationInformation_r9.otdoa_ProvideLocationInformation)) {
      if (isvalue (v_LppMessage.lpp_MessageBody.c1.provideLocationInformation.criticalExtensions.c1.provideLocationInformation_r9.otdoa_ProvideLocationInformation.otdoa_Error)) {
        f_EUTRA_SetVerdictFailOrInconc (__FILE__, __LINE__, "OTDOA Location Information present but reported error");
      }
      else {
        f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Test Case 7.3.4.4 Step 5");
      }
    }
    else {
      f_EUTRA_SetVerdictFailOrInconc (__FILE__, __LINE__, "OTDOA Location Information not present");
    }
    //@siclog "Step 5a" siclog@
    //SS sends LPP ACKNOWLEDGEMENT if requested by the UE at step 5
    f_POS_SendLppAck (eutra_Cell1,
                      tsc_RoutingId,
                      v_LppMessage);
    // Postamble
    f_EUTRA_TestBody_Set(false);
    f_EUTRA_Postamble(eutra_Cell1, E2_CONNECTED);//@sic R5-103875 sic@
  };
6. Execution Log Files

6.1 Qualcomm MDM 8960 UE 

The Qualcomm MDM 8960 UE passed this test case on R&S CMW500 with NAS and AS Snow 3G integrity and ciphering algorithm in LTE FDD band 1. The documentation below is enclosed as evidence of the successful test case run [1]:

· Test Case Execution log files: 
tc_7_3_4_4_5s_LTE_UE_Positioning_Qualcomm_FDD.log   
(Note: PICS/PIXIT settings are captured at the beginning of the TLI log)

In the log file (in .txt format) the complete test case execution can be seen. All TLI events are presented and message contents are fully decoded and can be verified. Preliminary verdicts and the final test case verdict can be seen in the log file.
6.2 Renesas Mobile MP5232-1 UE 

The Renesas Mobile MP5232-1 UE passed this test case on R&S CMW500 with NAS and AS Snow 3G integrity and ciphering algorithm in LTE FDD band 1. The documentation below is enclosed as evidence of the successful test case run [1]:

· Test Case Execution log files: 
tc_7_3_4_4_5s_LTE_UE_Positioning_Renesas_FDD.log   
(Note: PICS/PIXIT settings are captured at the beginning of the TLI log)

In the log file (in .txt format) the complete test case execution can be seen. All TLI events are presented and message contents are fully decoded and can be verified. Preliminary verdicts and the final test case verdict can be seen in the log file.

References

	[1]
	R5s130289:       Supporting information for agreement of EUTRA positioning TC 7.3.4.4_5s. 
                        This archive comprises:

                         -  text format execution log files.

